> Soluzioni del compito di Calcolo Numerico del 26 luglio 2004

> Enrico Bertolazzi

Esercizio 1

```
> restart:
  with(LinearAlgebra) :
```

Funzione di partenza

```
> f := x -> x^2 - x*cos(2*x) ;
```

$$f := x \to x^2 - x\cos(2\,x)$$

Derivata prima

```
> df := D(f);
```

$$df := x \to 2\,x - \cos(2\,x) + 2\,x\sin(2\,x)$$

Metodo di Newton

```
> Newton := unapply( factor(x - f(x) / df(x)), x);
```

$$Newton := x \to -\frac{x^2\,(1 + 2\sin(2\,x))}{-2\,x + \cos(2\,x) - 2\,x\sin(2\,x)}$$

Metodo di Newton a partire da x0 = 1:

```
> x0 := 1:
  x1 := evalf(Newton(x0),6):
  x2 := evalf(Newton(x1),6):
  x3 := evalf(Newton(x2),6):
  x4 := evalf(Newton(x3),6):
  <x0,x1,x2,x3,x4>;
```

$$\begin{bmatrix} 1 \\ 0.665587 \\ 0.546174 \\ 0.516923 \\ 0.514940 \end{bmatrix}$$

Metodo delle secanti

```
> Secanti := unapply( factor( (y*f(x)-x*f(y)) / (f(x)-f(y))), x, y);
```

$$Secanti := (x, y) \to \frac{y\,x\,(-x + \cos(2\,x) + y - \cos(2\,y))}{-x^2 + x\cos(2\,x) + y^2 - y\cos(2\,y)}$$

Metodo delle secanti a partire da x0 = 0 x1 =2 ;

```
> x0 := 0 :
  x1 := 2 :
  x2 := evalf(Secanti(x0,x1),4):
  x3 := evalf(Secanti(x1,x2),4):
  x4 := evalf(Secanti(x2,x3),4):
  <x0,x1,x2,x3,x4>;
```

$$\begin{bmatrix} 0 \\ 2 \\ 0. \\ 0. \\ \text{Float(undefined)} \end{bmatrix}$$

Esercizio 2

```
> restart:
  with(LinearAlgebra) :
```

```
> psol := x -> 1-x-16*x^3;
```

$$psol := x \rightarrow 1 - x - 16\,x^3$$

Punti di interpolazione:

```
> X := [-1,0,1,2,3];
  Y := [seq(psol(X[i]),i=1..5)];
```

$$X := [-1, 0, 1, 2, 3]$$

$$Y := [18, 1, -16, -129, -434]$$

Polinomio interpolante:

```
> interp( X, Y, 'z');
```

$$-16\,z^3 - z + 1$$

Costruzione delle differenze divise di ordine 0

```
> f1 := Y[1];
  f2 := Y[2];
  f3 := Y[3];
  f4 := Y[4];
  f5 := Y[5];
```

$$f1 := 18$$

$$f2 := 1$$

$$f3 := -16$$

$$f4 := -129$$

$$f5 := -434$$

Differenze divise

```
> f12 := (f2-f1)/(X[2]-X[1]);
  f23 := (f3-f2)/(X[3]-X[2]);
  f34 := (f4-f3)/(X[4]-X[3]);
  f45 := (f5-f4)/(X[5]-X[4]);
```

$$f12 := -17$$

$$f23 := -17$$

$$f34 := -113$$

$$f45 := -305$$

Differenze divise seconde

```
> f123 := (f23-f12)/(X[3]-X[1]);
  f234 := (f34-f23)/(X[4]-X[2]);
  f345 := (f45-f34)/(X[5]-X[3]);
```

$$f123 := 0$$

$$f234 := -48$$

$$f345 := -96$$

Differenze divise terze

```
> f1234 := (f234-f123)/(X[4]-X[1]);
  f2345 := (f345-f234)/(X[5]-X[2]);
```

$$f1234 := -16$$

$$f2345 := -16$$

Differenze divise quarte

```
> f12345 := (f2345-f1234)/(X[5]-X[1]);
```

$$f12345 := 0$$

Polinomi della base

```
> w0 := 1 ;
  w1 := x-X[1] ;
  w2 := expand(w1 * ( x - X[2])) ;
  w3 := expand(w2 * ( x - X[3])) ;
  w4 := expand(w3 * ( x - X[4])) ;
```

$$w0 := 1$$

$$w1 := x + 1$$

$$w2 := x^2 + x$$

$$w3 := x^3 - x$$

$$w4 := x^4 - 2x^3 - x^2 + 2x$$

Polinomio interpolante

```
> p := f1*w0 + f12 * w1 + f123 * w2 + f1234 * w3 + f12345 * w4 ;
```

$$p := 1 - x - 16\,x^3$$

Esercizio 3

```
> restart:
  with(Student[Calculus1]):
```

Funzione da integrare:

```
> f := x -> sin(3*x)-x*cos(2*x) ;
```

$$f := x \rightarrow \sin(3\,x) - x\cos(2\,x)$$

Integrale esatto

```
> a,b := -2,1:
  int(f(x),x=a..b);
  esatto := evalf(%);
```

$$\frac{1}{3}\cos(6) + \frac{1}{4}\cos(4) + \sin(4) - \frac{1}{3}\cos(3) - \frac{1}{4}\cos(2) - \frac{1}{2}\sin(2)$$

$$esatto := -0.6207711438$$

Derivata seconda della funzione

```
> ddf := (D@D)(f) ;
```

$$ddf := x \rightarrow -9\sin(3\,x) + 4\sin(2\,x) + 4\,x\cos(2\,x)$$

Possibile costante che maggiore il modulo della derivata seconda

```
> M := 9 + 4 + 8 ;
```

$$M := 21$$

Verifica grafica:

```
> plot(abs(ddf(x)),x=a..b) ;
```



Calcolo in numero minimo stimato di intervalli

```
> err := 10^(-3) = (b-a)^3/(12*n^2)*M ;
```

$$err := \frac{1}{1000} = \frac{189}{4\,n^2}$$

```
> isolate(err,n^2) ;
```

$$n^2 = 47250$$

```
> evalf(sqrt(rhs(%)));
```

$$217.3706512$$

Intervalli stimati: 218

```
> ApproximateInt(f(x), x=a..b, method = trapezoid, output = plot,
  partition = 6);
```

An Approximation of the Integral of
f(x) = sin(3*x)-x*cos(2*x)
on the Interval [-2, 1]
Using the Trapezoid Rule
Approximate Value: -.6207711437




```
> trap := evalf(ApproximateInt(f(x), x=a..b, method = trapezoid,
  partition = 6));
```

trap := -0.6500139605

```
> trap - esatto ;
```

-0.0292428167

Integrale con 218 intervalli

```
> evalf(ApproximateInt(f(x), x=a..b, method = trapezoid, partition =
  218));
```

-0.6207907480

Esercizio 4

```
> restart:
  with(LinearAlgebra) :
> sol := x -> 1-x-x^2;
```

$sol := x \to 1 - x - x^2$

Definizione del problema

```
> p := x -> x ;
  q := x -> -2 ;
  r :=
  unapply(expand(diff(sol(x),x,x)+p(x)*diff(sol(x),x)+q(x)*sol(x)),x) ;
  xa, xb := -2, 2 ;
  ya, yb := sol(xa), sol(xb) ;
```

$$p := x \to x$$

$$q := x \to -2$$

$$r := x \to -4 + x$$

$$xa, xb := -2, 2$$

$$ya, yb := -1, -5$$

Differenze finite:

```
> n := 4 ;
  h := (xb-xa) / n ;
  x[0] := xa ;
  x[1] := xa + h ;
  x[2] := xa + 2*h ;
  x[3] := xa + 3*h ;
  x[4] := xa + 4*h ;
```

$$n := 4$$

$$h := 1$$

$$x_0 := -2$$

$$x_1 := -1$$

$$x_2 := 0$$

$$x_3 := 1$$

$$x_4 := 2$$

```
> eq := k -> (y[k+1]-2*y[k]+y[k-1])/h^2 +
             p(x[k]) * (y[k+1]-y[k-1])/(2*h) +
             q(x[k]) * y[k] - r(x[k]) ;
```

$$eq := k \to \frac{y_{k+1} - 2\, y_k + y_{k-1}}{h^2} + \frac{p(x_k)\,(y_{k+1} - y_{k-1})}{2\, h} + q(x_k)\, y_k - r(x_k)$$

Equazioni risultanti

```
> eq1 := eq(1) ;
  eq2 := eq(2) ;
  eq3 := eq(3) ;
```

$$eq1 := \frac{1}{2}\, y_2 - 4\, y_1 + \frac{3}{2}\, y_0 + 5$$

$$eq2 := y_3 - 4\, y_2 + y_1 + 4$$

$$eq3 := \frac{3}{2}\, y_4 - 4\, y_3 + \frac{1}{2}\, y_2 + 3$$

Estraggo il sistema lineare dalle equazioni

```
> A := linalg[genmatrix]([eq1,eq2,eq3],[y[1],y[2],y[3]],'b'):
  A := convert(A,Matrix);
```

$$A := \begin{bmatrix} -4 & \frac{1}{2} & 0 \\ 1 & -4 & 1 \\ 0 & \frac{1}{2} & -4 \end{bmatrix}$$

```
> b := Transpose(convert(b,Vector));
```

$$b := \begin{bmatrix} -\frac{3}{2}\, y_0 - 5 \\ -4 \\ -\frac{3}{2}\, y_4 - 3 \end{bmatrix}$$

Sostituisco le condizioni al contorno

```
> b := subs(y[0]=ya, y[4]=yb, b) ;
```

$$b := \begin{bmatrix} \frac{-7}{2} \\ -4 \\ \frac{9}{2} \end{bmatrix}$$

Risolvo il sistema lineare

```
> LinearSolve(A,b) ;
```

$$\begin{bmatrix} 1 \\ 1 \\ -1 \end{bmatrix}$$

Esercizio 5

```
> restart:
  with(LinearAlgebra) :
```

Tabella dei punti

```
> X := [-2,-2,-1,0,1,2,2,3,2,1];
  Y := [-1,-1,0,1,3,2,4,-2,3,1];
```

$$X := [-2, -2, -1, 0, 1, 2, 2, 3, 2, 1]$$

$$Y := [-1, -1, 0, 1, 3, 2, 4, -2, 3, 1]$$

```
> n := nops(X) ;
```

$$n := 10$$

```
> SX   := add(X[i],i=1..n) ;
  SX2  := add(X[i]^2,i=1..n) ;
  SX3  := add(X[i]^3,i=1..n) ;
  SX4  := add(X[i]^4,i=1..n) ;
  SY   := add(Y[i],i=1..n) ;
  SXY  := add(X[i]*Y[i],i=1..n) ;
  SX2Y := add(X[i]^2*Y[i],i=1..n) ;
```

$$SX := 6$$

$$SX2 := 32$$

$$SX3 := 36$$

$$SX4 := 164$$

$$SY := 10$$

$$SXY := 20$$

$$SX2Y := 14$$

```
> A := <<n,SX,SX2>|<SX,SX2,SX3>|<SX2,SX3,SX4>>;
```

$$A := \begin{bmatrix} 10 & 6 & 32 \\ 6 & 32 & 36 \\ 32 & 36 & 164 \end{bmatrix}$$

```
> b := <SY,SXY,SX2Y> ;
```

$$b := \begin{bmatrix} 10 \\ 20 \\ 14 \end{bmatrix}$$

```
> res := LinearSolve(A,b) ;
  evalf(res);
```

$$res := \begin{bmatrix} \frac{1973}{917} \\ \frac{104}{131} \\ \frac{-933}{1834} \end{bmatrix}$$

$$\begin{bmatrix} 2.151581243 \\ 0.7938931298 \\ -0.5087241003 \end{bmatrix}$$

Soluzione

```
> x := 'x' :
  p := 'p' :
```

```
> p := res[1]+ x * res[2] + x^2 * res[3] ;
```

$$p := \frac{1973}{917} + \frac{104}{131} x - \frac{933}{1834} x^2$$

```
> evalf(p,4) ;
```

$$2.152 + 0.7939 x - 0.5087 x^2$$